' Class List 4 - © 1993-2002 Aiman Hanna

' This program explores multidimensional arrays. 

' Key Points:

'       1) Multidimensional Arrays

'       2) Use LBound & UBound for multidimensional arrays

‘ Notes: A multidimensional arrays is created in a similar way to one-dimensional array, except for that 

‘ there are two values inside the parentheses, for example: Dim clarr (5, 7) as Long, will create an array

‘ (looks like a table) with 5 rows and each row has 7 columns, assuming that Option Base 1 is indicated.

‘ (Again, if Option Base is set to the default (zero), this will create an array of 6 rows and 8 columns). 

‘ To find the lower and upper bound of the rows (first dimension), we use LBound(clarr) and

‘ UBound(clarr). To get the lower and upper bounds of the columns (the second dimension), we use 

‘ LBound(clarr, 2) and UBound(clarr, 2).; the value 2 here simply indicates dimension number 2. 

‘ With VB, a programmer can create arrays with at least 60 array dimensions. However, the use of 

‘ 3-dimensional arrays is minimal, and the use of more than 3 dimensions is indeed rare. 

Option Base 1

Option Explicit

Dim clarr(5, 7) As Long

Private Sub cmdGenerateClassLists_Click()

    Dim str As String

    Dim i As Integer, j As Integer

    ' Clear the form

    Cls

    Call Randomize

    ' Initialize the first column of the array to the course number

    clarr(1, 1) = 201

    clarr(2, 1) = 228

    clarr(3, 1) = 249

    clarr(4, 1) = 346

    clarr(5, 1) = 352

    'Print six empty lines just to go below the Generate Class List button

    For i = 1 To 6

        Print

    Next i

    ' Now fill the classes with IDs. This is the second dimension of the array

    ‘ Do not insert Ids in the first column since it has the course number

    For i = LBound(clarr) To UBound(clarr)

        str = "Comp" & clarr(i, 1) & vbTab

        For j = LBound(clarr, 2) + 1 To UBound(clarr, 2)

            clarr(i, j) = 10000000 + Int(10000000 * Rnd())

            str = str & clarr(i, j) & vbTab

        Next j

        Print

        Print str

        Print

        ' print a horizontal line

        Print String$(100, "_")

    Next i

    ' Now the class lists have been created, allow the search

    txtSearchID.BackColor = &H8000000E

    txtSearchID.Enabled = True

    cmdSearch.Enabled = True

End Sub

Private Sub cmdSearch_Click()

    Dim i As Integer, j As Integer

    Dim id As Long

    ' The program should verify that there is an input to search for. For simplicity,

    ' this is omitted from this program. See the "ArrayOperations1" program for an example

    ' of such verifications

    lblSearchMsg.Caption = ""

    id = txtSearchID.Text

    For i = LBound(clarr) To UBound(clarr)

        For j = LBound(clarr, 2) + 1 To UBound(clarr, 2)

            If clarr(i, j) = id Then

                lblSearchMsg.Caption = "The ID was found at index (" & i & " , " & j & ")." _

                                        & " This is the " & "Comp" & clarr(i, 1) & " class."

                Exit For

            End If

        Next j

        If j <= UBound(clarr, 2) Then   ' This indicates that the inner loop was exited

                Exit For

        End If

    Next i

    If i > UBound(clarr) Then

        lblSearchMsg.Caption = "The ID was not found in any class."

    End If

End Sub 

Private Sub cmdExit_Click()

    End

End Sub

